
			Parrilla de juegos

		

	
		
			Hola Programas

			Diseñar juegos en el ZX Spectrum ahora es más fácil gracias al Multi-Platform Arcade Game Designer.

			Este editor se encarga de organizar los bloques de rutinas y datos de los que constará nuestro Juego Arcade.

			Es la evolución del Arcade Game Designer, un editor "nativo" para el ZX Spectrum. Si tienes un ZX Spectrum en casa, te propongo que pruebes a usarlo; con la dificultad añadida al usar las antiguas herramientas para guardar tu trabajo: el "cassette".

			Como alternativa al ZX Spectrum y para el desarrollo ágil de nuestro juego, usaremos la emulación del ZX Specrum 48k y 128k.

			La experiencia de la emulación es igualmente evocadora y nos transporta a los "antiguos" tiempos en los que alucinábamos con la magia de la informática y jugar con estas nuevas máquinas nos descubría mil maravillas.

			Podremos guardar múltiples cintas (.tap) con múltiples grabaciones (Program, screen, bin, data) y organizarlas en nuestro ordenador con la eficacia de sistemas de este siglo.

			Además la emulación ha llegado a todo tipo de dispositivos, con lo que podrás jugar incluso en la tele de tu salón.

			Te dejo aquí los enlaces para que te instales dos herramientas. Visita la web y el tutorial del MPAGD para saber más.

			MPAGD

			Instala el MPAGD (Multi-Platform Arcade Game Designer).

			https://jonathan-cauldwell.itch.io/multi-platform -arcade-game-designer

			Para este tutorial usamos el paquete V0.7.10.zip. Descomprime en una carpeta con un nombre y una ruta que no contengan espacios (por ejemplo MPAGD710)

			[image: ]

			En la imagen: el editor de sprites con nuestro player, frame 0.

			EDITOR DE TEXTO

			Es obvio que usaremos un editor. Para nuestro proyecto es de gran utilidad NOTEPAD++. No es necesario para seguir el tutorial; pero facilita mucho la labor de edición.

			Puedes descargarlo en notepad-plus-plus.org.

			https://notepad-plus-plus.org

			Para usarlo necesitas seguir estos pasos:

			1.Descarga e instala el producto NOTEPAD++

			2.Prepárate para Editar las variables de entorno de sistema de Windows 10. Busca algún tutorial.

			3.Añade una nueva ruta PATH en el editor de variables de entorno de sistema, escoge el directorio de instalación del NOTEPAD++.

			4.Ahora busca dentro del directorio de instalación del MPAGD el archivo de configuración mpagd.ini y edítalo para que quede así:

			
				
					
				
				
					
							
							background_colour = 001f2f

							script_editor = Notepad++.exe

							compiler_zx = build-zx.bat

							compiler_timex = build-tmx.bat

							compiler_cpc = build-cpc.bat

							compiler_atom_mono = build-atom.bat

							compiler_dragon_mono = build-dragon.bat

							compiler_coco_mono = build-coco.bat

							compiler_bbc = build-bbc.bat

							compiler_electron = build-electron.bat

							compiler_next = build_vz200.bat

							compiler_enterprise = build-ep.bat

							compiler_msx = build-msx.bat

							compiler_vz200_mono = build-vz200.bat

							compiler_atom_colour = build-atom-colour.bat

							compiler_dragon_colour = build-dragon-colour.bat

							compiler_coco_colour = build-coco-colour.bat

							compiler_vz200_colour = build-vz200-colour.bat

							export_zx = export-zx.bat

							export_timex = export-tmx.bat

						
					

				
			

			Hemos cambiado el valor de script_editor para usar nuestro editor favorito. Guarda el archivo.

			Estamos listos. "GET READY!"

		

	
		
			Proyecto Pacman

			Comenzamos nuestro proyecto Pacman creando una carpeta con ese nombre en el lugar donde tu prefieras.

			Arrancamos nuestro editor MPAGD. Para ello ejecutamos winagd.exe (yo le he creado un acceso directo en el escritorio). Lo primero que haremos es guardar nuestro nuestro proyecto vacío. Seleccionamos la opción "Save proyect" del menú "File", lo guardamos en la carpeta que hemos creado y con el nombre de proyecto Pacman. Cerramos el MPAGD (opción "Exit" del menú "File").

			Volvemos a cargar el proyecto. Seleccionamos opción "Load" del menú "File" y buscamos Pacman.apj en nustra carpeta Pacman.

			Por último pulsa "F5" (opción "Build" del menú "File") y deberá aparecer un juego vacío:

			[image: ]

			Dentro de la carpeta Pacman se habrán creado una serie de archivos correspondientes a los scripts de código que vamos a programar. Estos scripts se ejecutan en determinados eventos del juego.

			Otro archivo es el script del editor, con extensión .agd donde se describen los bloques de datos y secciones del juego: pantallas, mapas, Sprites, eventos, sonido, fuentes, paleta de color, etc.

			Todos estos archivos los maneja el editor. No los toques por ahora.

			Si no cierras el emulador, no podrás controlar de nuevo el editor.

			DEFINEWINDOW

			Este es el primer bloque que vamos a definir hoy: "la ventana de juegos”.

			Pac-man se jugaba en una "TV de tubo” dispuesta en vertical (busca CTR TV en YouTube). Pero el escenario del ZX Spectrum es horizontal. Está dispuesto en celdas ("caracteres") de 8x8 pixeles dispuestas en 21 filas y 31 columnas.

			Selecciona la opción "Window size" en el menú "Editor”. Aparece una parrilla de juego en la que está representada la pantalla completa del ZX Spectrum.

			Procede de la siguiente manera. Haz "clic" en la ventana de juego, quedará seleccionada en rojo. Selecciona con otro "clic" la esquina superior izquierda en la casilla 1,1 de la parrilla y el último "clic" es para la esquina inferior derecha en la casilla 20,30 de la parrilla. Hazlo en este orden, así queda fijada la "ventana del juego".

			Otras cosas que colocar. De la misma manera coloca el "marcador de puntos” a la izquierda, el de "récord” a la derecha y las "vidas” en el centro tal como se ilustra en la figura. Los botones que aparecen a la derecha del editor añaden o eliminan estas características del juego. Coloca todo esto como en la figura (necesitarás recolocar la ventana de juego para poder ver y seleccionar el marcador, récord y vidas):

			[image: ]

			Guarda el proyecto de nuevo (opción "Save proyect" del menú "File").

			Hecho!! Pero, ¿Qué hemos definido con este pequeño trabajo? Pues bien:

			Hemos tomado varias decisiones de diseño en cuanto a la colocación del tablero de juego y marcadores.

			Hemos dado a nuestro juego una motivación: obtener un "SCORE”, algo que ganar, pero también algo que perder "LIVES”, nada más y nada menos. Además podemos presumir de nuestra maestría a través de nuestro "RECORD”.

			Estas son las motivaciones de los Arcade. Nos invitan a pasar horas y horas jugando. Pac-man ofrece otras satisfacciones que lo hacen especial. Cada juego, en aquellos tiempos, ofrecía mediante expectativas, deseos y trucos sencillos un entretenimiento que enganchaba.

			Descubrir y analizar los elementos de estos juegos resultará útil para diseñar tus próximos juegos.

			El editor ha hecho su trabajo también.

			Ha definido los valores de algunas constantes y variables en el juego: WINDOWWID, WINDOWHGT, WINDOWLFT, WINDOWTOP, LEFTEDGE, RIGHTEDGE, TOPEDGE, BOTTOMEDGE, LIVES, SCORE y HISCORE. Todas estas variables las gestionará el EngineZX, motor que trabaja entre bastidores, nosotros no nos preocupamos de esto.

			SCRIPT GENERATOR

			Selecciona la opción "Scripts Generator" del menú "Tools".

			Esta herramienta es la más potente del MPAGD. Vamos a usarla para generar código a partir de las posiciones y elementos que hemos seleccionado en el apartado anterior "Window size".

			[image: ]

			Deja todos los botones de la segunda columna en la opción [existing] excepto el último. Selecciona Auto generate para la sección "Other events". Pulsa el botón "Generate Scripts" con seguridad.

			Ahora revisa en el menú "Events" y selecciona el evento "Game initialisation”. El editor ha escrito lo siguiente:

			
				
					
				
				
					
							
							EVENT GAMEINIT

							LET LIVES = 3; la variable global LIVES es 3

							AT 22 16; sitúa el cursor en pantalla en celda [16,22]

							DISPLAY LIVES ; pinta el valor de la variable LIVES

							AT 22 2

							SHOWSCORE; pinta los dígitos del SCORE.

							AT 22 24

							SHOWHIGH; pinta los dígitos del HIGHSCORE

						
					

				
			

			El editor ha escrito la posición en pantalla de nuestros marcadores y nos concede 3 vidas!

			Si te fijas nombramos primero la coordenada vertical y después la horizontal si usamos el comando AT. Comentar las líneas es una buena práctica para documentar el proyecto y entender el código.

			Guarda el proyecto de nuevo (opción "Save proyect" del menú "File").

			Puedes probar ahora. "Build" es una opción del menú "File" del MPAGD, puedes pulsar F5 también. Vigila la ventana de comandos por si hubiera algún error.

			[image: ]

			error: Label not found: nmedat

			No te preocupes. Este error solo se produce cuando no se ha definido ningún enemigo en la pantalla. Pero aún no hemos empezado a trabajar. En los siguientes capítulos será lo primero que haremos. Cuando esté todo correcto y pulses F5 aparecerá la siguiente pantalla.

			[image: ]

			Otra cosa que hará el "Script Generator" es escribir el menú de inicio que será lo primero que aparezca al arrancar el juego y la única vez que lo haga. Aquí podemos escoger el tipo de control para usar en el juego. También puedes crear tu propio menú.

			[image: ]

			En la imagen: la nueva disposición de ventana y marcadores del juego

			Déjalo por hoy. Marca un día a la semana o al mes para preocuparte de este proyecto. Esto es un consejo para disfrutar de este hobby un largo tiempo. Merece la pena.

		

	
		
			El laberinto

			"... creé un laberinto y puse la comida en él. Entonces, quien juega el juego tendría la obligación de moverse a través del laberinto".

			Toru Iwatani, El dosier fantasma.

			Esta decisión la toma para ofrecerle al jugador o jugadora una motivación clara y sencilla. Instala una lógica en el tablero de juego. Pac-man se debe mover por los caminos del laberinto, con dos salidas y entradas el los extremos. Los perseguidores, los fantasmas, usan la misma lógica. Ahora todos sabemos las reglas.

			[image: ]

			figura: "laberinto"

			Elegimos esta disposición por las características del MPAGD que disponen los gráficos del decorado en BLOQUES de 8x8 pixeles. Empezamos a contar en orden de izquierda a derecha y de arriba a abajo de manera que la celda más arriba a la izquierda es la 0,0.

			Muy bien. ¿Cómo hacemos esto?

			Blocks

			Abrimos el editor MPAGD. Abrimos el proyecto Pacman (esto lo repetiremos al inicio de cada capítulo). En el menú "Editor", seleccionamos "Blocks".

			Vamos a dibujar uno a uno los bloques o azulejos que configuran nuestro laberinto. Para ello puedes ayudarte de unas teclas rápidas que encontrarás si pulsas "F1" o en el manual del editor.

			Vamos con el primero:

			[image: ]

			Block 000 EMPTYBLOCK. Este bloque número 000 y tipo EMPTYBLOCK se suele reservar para el fondo, suele ser negro y estar vacío ("empty" en inglés), sin pixeles. No lo vamos a editar (asegúrate de que permanezca como EMPTYBLOCK).

			Vamos a crear uno nuevo. Pulsa "x" (busca la tecla rápida en el menú de ayuda o pulsa "F1").

			[image: ]

			Figura: Block 001 WALLBLOCK. Dibuja los pixeles en el área de edición (arriba a la izquierda).

			Block 001 WALLBLOCK. Ahora sí. Edita el bloque numero 001 cambiando el tipo de EMPTYBLOCK para poner "WALLBLOCK" (muro). Para cambiar el tipo puedes hacer clic en el botón situado debajo de los botones de INK (tinta) y PAPPER (papel). Escoge el color de tinta que te guste.

			Crea el resto de bloques que incluye la lista.

			Para cada nuevo bloque repite lo siguiente. Crea bloque (Pulsa "x"), edita el gráfico y modifica el tipo dejándolo en WALLBLOCK. Algunos son simétricos en el eje vertical y otros en el eje horizontal, de manera que puedes copiarlos y reflejarlos para ahorrar trabajo (ayúdate de las teclas rápidas que encontrarás si pulsas "F1").

			Aquí tienes la lista:

			[image: ]

			El tipo WALLBLOCK o muro, es un adorno, bloque o azulejo que se pinta en pantalla y que no se puede atravesar (como ves el propio tablero aporta lógica al juego).

			[image: ]

			[image: ]

			[image: ]

			[image: ]

			[image: ]

			Block 009 COLLECTABLE. Este tipo especial de bloque reacciona cuando colisiona si activamos el mecanismo GETBLOCKS: por un lado se borra en pantalla y además dispara el evento COLLECTBLOCK donde nosotros podemos intervenir mediante un script. Los usamos para ¡comer!

			[image: ]

			[image: ]

			[image: ]

			Block 012 LADDERBLOCK es otro tipo especial de bloque que puede atravesarse o sobrepasarse, además también puede ser detectado mediante el método LADDERBELOW y LADDERABOVE. Será la salida de la "cueva fantasmal".

			[image: ]

			[image: ]

			[image: ]

			[image: ]

			[image: ]

			Block 017 WALLBLOCK lo usaremos para dibujar un camino invisible en la pantallas de INTERMISSION y en la PRESENTACION.

			[image: ]

			Block 018 EMPTYBLOCK (lo dejamos por defecto EMPTYBLOCK ) lo usamos de icono de las VIDAS.

			[image: ]

			Nos deberá quedar este "Tile set" o conjunto de azulejos para almacenarse en un bloque concreto de memoria en nuestro programa.

			Guarda el proyecto.

			Screens

			Ahora sólo queda dibujar la pantalla del "Laberinto" (igual que en la figura "laberinto"). Pinta siguiendo las referencias y distancias que te proporcionan los puntos amarillos.

			Selecciona la opción "Screens" del menú "Editor". Aquí pinta con ayuda de teclas rápidas y el ratón nuestra pantalla (las teclas rápidas las encontrarás si pulsas "F1" o en el manual del editor).

			Guarda el proyecto.

			Bien hecho. Has definido la lógica de tablero: Bloques infranqueables, puntos de luz para comer, barreras fantasmales, túneles que nos tele-transportan. Un tablero de juego lleno de lucecitas y color que pronto se llenará de animación y estrategia.

			El próximo día colocamos a Puck-Man.

			"Con un nombre y un diseño básico, Iwatani estaba listo para comenzar a trabajar. El equipo que Namco asignó a Iwatani para darle vida a Puck-Man incluía un programador (Shigeo Funaki), un ingeniero de hardware, un equipo de diseñadores y un compositor de música (Toshio Kai)."

			Jamey Pittman, El dossier de Pac-Man

		

	
		
			Pac-man

			"Los japoneses tienen una jerga, paku paku, que usan para describir el movimiento de la boca abriéndose y cerrándose mientras se come. El nombre Puck-Man vino de esa palabra ".

			Toru Iwatani

			Sprites

			Todos sabemos cómo es Pacman y cómo se comporta. Esto es una ventaja para nosotros. El reto es usar el MPAGD para conseguir cumplir los requerimientos de Pacman como si fueran una orden de nuestro diseñador jefe. Imagina que estas trabajando en una empresa de videojuegos de 1985 pero con las ventajas que nos dan las herramientas de edición de texto y gráficas del siglo 21. Todo lo que tienes es un manual secreto: "El dosier fantasma" que será tu única guía para ayudarte a lograrlo.

			¿Qué hace Pacman?

			Es una imagen que se mueve en cuatro direcciones en el tablero de juego, mientras se mueve abre y cierra la boca. Un buen comienzo.

			Resolvamos esto. Abrimos el MPAGD. Abrimos nuestro proyecto Pacman. En el menú "Editor" seleccionamos "Sprites".

			[image: ]

			Sprite 000 frame 000

			Dibuja pixel a pixel a nuestra pizza amarilla. Toma referencias contando los pixeles para que te salga idéntico al de la figura.

			Ahora nuestro Pacman puede tomar la apariencia de la imagen Sprite 000 frame 000.

			Lo siguiente es animar a Pacman. Creamos el frame 001. Pulsamos "i" (insertar nuevo frame) y dibujamos la siguiente:

			[image: ]

			Sprite 000 frame 001

			Uno más. Volvemos a pulsar "i" para insertar un frame más:

			[image: ]

			Sprite 000 frame 002

			Bien tenemos una pequeña animación de tres frames, suficiente ya lo verás.

			(Guarda el proyecto y sitúate en el editor de Sprites en el último Sprite que hayas dibujado, es decir Sprite 000 de nuevo).

			Toca mirar a la izquierda. Pulsamos "x" con lo que vamos a crear un nuevo Sprite. Puedes dibujar de nuevo o copiar cada frame del modelo anterior y hacer reflexión para que mire a la izquierda esta vez (usa el menú de ayuda pulsando "F1" para localizar los atajos de teclado). Debería quedar como en las siguientes imágenes:

			[image: ]

			Sprite 001 frame 000

			[image: ]

			Sprite 001 frame 001

			[image: ]

			Sprite 001 frame 002

			Usando el mismo método anterior dibuja Pacman mirando al cielo y al suelo. Usa el mismo orden para las imágenes, es importante. No dudes usar la ayuda del menú pulsando "F1". También tienes ayuda en la documentación del programa.

			[image: ]

			Sprite 002 frame 000

			[image: ]

			Sprite 002 frame 001

			[image: ]

			Sprite 002 frame 002

			[image: ]

			Sprite 003 frame 000

			[image: ]

			Sprite 003 frame 001

			[image: ]

			Sprite 003 frame 002

			Guarda el proyecto.

			Hasta el momento hemos llenado la memoria de nuestro programa de bits y mas bits de datos gráficos.

			Estas imágenes se organizarán en memoria en bloques de 16x16 para que las podamos reproducir en pantalla. Podemos hacerlo de forma poderosa ya que cuando construyes un personaje puede tomar la apariencia de cualquiera de ellas, la animación corre a cuenta del EngineZX.

			"Puck-Man pasó por la mayoría de los cambios: el gabinete se modificó ligeramente, cambiando el color de blanco a amarillo brillante para que se destaque en la sala de juegos. La obra de arte detallada y multicolor del "arcade cabinet" fue reemplazada por una obra de arte de tres colores más barata de producir que ilustra una representación icónica de Puck-Man (ahora dibujado con ojos y pies) y un fantasma azul".

			Jamey Pittman, El dossier de Pac-Man

		

	
		
			Esto es un juego

			Estamos nerviosos delante de nuestro jefe de desarrollo. Está mirando nuestros diseños. Mira el laberinto y a Pacman. Después de un silencio dice: "perfecto, me gusta el color, va a destacar en la sala de juegos; pero ¿Qué hace Pacman?"

			Tu le dices: "La premisa de Pac-Man es deliciosamente simple: con un joystick de cuatro vías, el jugador o jugadora guía a Pac-Man hacia arriba, hacia abajo, hacia la izquierda y hacia la derecha a través de un laberinto lleno de puntos para que los engulla".

			Otro silencio. "¿Dónde puedo ver eso?"

			SPRITE POSITIONS

			No hay problema, se lo vamos a mostrar.

			Abrimos nuestro MPAGD, cargamos nuestro proyecto Pacman. Escogemos la opción "Sprite Positions" del menú "Editor".

			[image: ]

			En la imagen: a la derecha botón para cambiar tipo, abajo iconos para seleccionar Sprite.

			Primer clic: Para AÑADIR a Pacman al "Laberinto" seleccionamos el icono correspondiente. Con cada "clic" estarás añadiendo un Sprite, ten cuidado

			Segundo clic: EL botón derecho muestra el Sprite seleccionado y su tipo. Debes dejar seleccionado el tipo "PLAYER RESPAWN".

			Tercer clic: En el Laberinto, en la "posición de salida", "GET READY!", que es la celda [14,11] (columna 14,fila 11).

			Ya tenemos la Instancia, aparición, "spawn" o reaparición, "respawn" definida en el EngineZX. Cada vez que el Player MUERE en esta pantalla, REAPARECE en esta posición.

			Ahora vamos a darle personalidad. Glotón compulsivo, rápido y decidido.

			Guarda el trabajo.

			EVENT PLAYER

			Escogemos la opción "Player" del menú "Events".

			[image: ]

			En la imagen: editor de texto preferido con el script abierto.

			Aquí leemos "EVENT PLAYER". Esta es una etiqueta para que nuestro EngineZX encuentre el script y lo ejecute.

			Las cosas en el juego ocurren en un orden concreto y se repiten en un bucle infinito en lo que se conoce como "GAME LOOP".

			"EVENT PLAYER" no es el primero de la lista de tareas del "GAME LOOP", hay cosas que ocurren antes y después. Esta arquitectura nos permite enfocar nuestra atención en departamentos concretos: ahora nos preocupa el control de Pacman.

			A continuación escribimos nuestro código, todo quedará así:

			
				
					
				
				
					
							
							EVENT PLAYER

							; Pacman

							SPRITEINK 14

							; Control

							IF KEY UP

							IF CANGOUP

							DIRECTION = UP

							ENDIF

							ENDIF

						
					

					
							
							IF KEY DOWN

							IF CANGODOWN

							DIRECTION = DOWN

							ENDIF

							ENDIF

							IF KEY LEFT

							IF CANGOLEFT

							DIRECTION = LEFT

							ENDIF

							ENDIF

							IF KEY RIGHT

							IF CANGORIGHT

							DIRECTION = RIGHT

							ENDIF

							ENDIF

							IF DIRECTION = UP

							IF CANGOUP

							SPRITEUP

							LET IMAGE = 2

							ANIMATE

							ENDIF

							ENDIF

							IF DIRECTION = DOWN

							IF CANGODOWN

							SPRITEDOWN

							LET IMAGE = 3

							ANIMATE

							ENDIF

							ENDIF

							IF DIRECTION = LEFT

							IF X <= LEFTEDGE

							LET X = RIGHTEDGE

							EXIT

							ELSE

							IF CANGOLEFT

							SPRITELEFT

							LET IMAGE = 1

							ANIMATE

							ENDIF

							ENDIF

							ENDIF

						
					

					
							
							IF DIRECTION = RIGHT

							IF X >= RIGHTEDGE

							LET X = LEFTEDGE

							EXIT

							ELSE

							IF CANGORIGHT

							SPRITERIGHT

							LET IMAGE = 0

							ANIMATE

							ENDIF

							ENDIF

							ENDIF

						
					

				
			

			Esto es el control de nuestro jugador. "A primera vista no entiendo mucho lo que significa este código" dices. "No te preocupes, yo te comento línea por línea para que todos podamos comprender y recordar el significado cuando volvamos a leerlo" dice el programador del equipo. Esto es muy buena práctica.

			Quedará así:

			
				
					
				
				
					
							
							EVENT PLAYER

							; Pacman

							SPRITEINK 14

							; Control

							; Selecciona la dirección en que movemos a Pacman.

							IF KEY UP	; si pulsamos la tecla "arriba"

							IF CANGOUP	; y si puede ir hacia arriba

							DIRECTION = UP	; entonces su dirección será "arriba"

							ENDIF

							ENDIF

							IF KEY DOWN

							IF CANGODOWN

							DIRECTION = DOWN

							ENDIF

							ENDIF

						
					

					
							
							IF KEY LEFT

							IF CANGOLEFT

							DIRECTION = LEFT

							ENDIF

							ENDIF

							IF KEY RIGHT

							IF CANGORIGHT

							DIRECTION = RIGHT

							ENDIF

							ENDIF

							; movemos a Pacman. Sección de movimiento

							IF DIRECTION = UP	; si su dirección es “arriba”

							IF CANGOUP	; y si puede ir hacia arriba

							SPRITEUP	; mueve arriba

							LET IMAGE = 2	; tu imagen es “mirando al cielo”

							ANIMATE	; ejecuta un frame de animación

							ENDIF

							ENDIF

						
					

					
							
							IF DIRECTION = DOWN

							IF CANGODOWN

							SPRITEDOWN

							LET IMAGE = 3

							ANIMATE

							ENDIF

							ENDIF

							IF DIRECTION = LEFT	; si su dirección es “izquierda”

							IF X <= LEFTEDGE	; y “X” supera el límite izquierdo de la pantalla “entra en el túnel del laberinto”

							LET X = RIGHTEDGE	; entonces colócate en el límite derecho “sal del túnel por el otro lado”

							EXIT	; salir del script

							ELSE

							IF CANGOLEFT

							SPRITELEFT

							LET IMAGE = 1

							ANIMATE

							ENDIF

							ENDIF

							ENDIF

						
					

					
							
							IF DIRECTION = RIGHT

							IF X >= RIGHTEDGE

							LET X = LEFTEDGE

							EXIT

							ELSE

							IF CANGORIGHT

							SPRITERIGHT

							LET IMAGE = 0

							ANIMATE

							ENDIF

							ENDIF

							ENDIF

						
					

				
			

			El programa deja que Pacman corra de manera compulsiva mientras no le indiquemos otra dirección o se encuentre con un obstáculo WALLBLOCK.

			Guarda el código desde tu editor de texto. Cierra el editor de texto.

			Guarda el proyecto.

			Prueba. Selecciona "Build" desde el menú "File". Atiende a las indicaciones de la ventana de comandos anexa al editor. Aquí puedes adivinar dónde ocurre un error.

			[image: ]

			en la figura: el menu inicial del juego.

			Selecciona con la tecla "1" el control con el teclado cuando aparezca este menú. Las teclas de control por defecto son "0","9","2","w". Lo puedes cambiar en la opción "Keyboard Controls" del menú "Editor".

			Ya es un pequeño juego, conducir a un "come cocos" por el "Laberinto", lo habéis logrado. Sólo nos queda completar el requerimiento de alimentar a Pacman con sus cocos y hacer crecer el "SCORE". Pero hoy tenéis una cita con Halloween. El jefe puede esperar. En una semana habremos acabado.

			Vamos a entender un poco más. Prueba este otro código:

			
				
					
				
				
					
							
							EVENT PLAYER

							; En otro juego

							SPRITEINK 5

							; Control

							IF KEY UP

							IF CANGOUP

							DIRECTION = UP

							SPRITEUP

							LET IMAGE = 2

							ANIMATE

							ENDIF

							ENDIF

							IF KEY DOWN

							IF CANGODOWN

							SPRITEDOWN

							LET IMAGE = 3

							ANIMATE

							LET DIRECTION = DOWN

							ENDIF

							ENDIF

							IF KEY LEFT

							IF X <= LEFTEDGE

							LET X = RIGHTEDGE

							EXIT

							ELSE

							IF CANGOLEFT

							SPRITELEFT

							LET IMAGE = 1

							ANIMATE

							LET DIRECTION = LEFT

							ENDIF

							ENDIF

							ENDIF

						
					

					
							
							IF KEY RIGHT

							IF X >= RIGHTEDGE

							LET X = LEFTEDGE

							EXIT

							ELSE

							IF CANGORIGHT

							SPRITERIGHT

							LET IMAGE = 0

							ANIMATE

							LET DIRECTION = RIGHT

							ENDIF

							ENDIF

							ENDIF

						
					

				
			

			Guarda el código desde tu editor de texto. Cierra el editor de texto.

			Prueba. Selecciona "Build" desde el menú "File". Atiende a las indicaciones de la ventana de comandos anexa al editor. Aquí puedes adivinar dónde ocurre un error. Un error muy común ocurre si dejas la última línea de tus scripts sin "end of line". Asegúrate de pulsar "enter" al final da esa última línea.

			La diferencia es que ahora tienes que pulsar la tecla para mover el jugador. Esto es porque el script se ejecuta en orden estricto de arriba a abajo, resolviendo cada uno de las líneas.

			Vuelve a colocar el código comentado de Pacman. Guarda el Script.

			Guarda el proyecto.

			Ahora es cuando nuestro programador presume de su trabajo y nos explica lo razonado, eficaz y ágil que es:

			CANGORIGHT, CANGOLEFT, CANGODOWN y CANGOUP nos ayudan a detectar la colisión con los bloques del "laberinto" tipo "muro".

			SPRITEINK colorea.

			SPRITERIGHT, SPRITELEFT, SPRITEDOWN y SPRITEUP mueven el Sprite 2 pixeles en las direcciones respectivas.

			ANIMATE ejecuta un frame de animación cada vez en un bucle infinito.

			RIGHTEDGE y LEFTEDGE los establecimos cuando definimos las dimensiones de nuestra pantalla.

			IMAGE es el índice que apunta a los gráficos (Sprites) que hemos editado y guardado en memoria. cuando cambiamos de dirección, cambiamos el gráfico Sprite.

			UP, DOWN, LEFT y RIGHT son constantes en el juego.

			Por último decir que IMAGE, Y, X y DIRECTION son propiedad de Pacman. Cada Sprite en juego tendrá su posición, su imagen, dirección y otras propiedades que veremos más adelante.

			Tenemos trabajo y juego por delante. Esto es un juego.

			"La premisa de Pac-Man es deliciosamente simple: con un joystick de cuatro vías, el jugador guía a Pac-Man hacia arriba, hacia abajo, hacia la izquierda y hacia la derecha a través de un laberinto lleno de puntos para que los engulla. Cuatro monstruos fantasmas también están en el laberinto y persiguen a nuestro héroe, tratando de capturarlo y matarlo".

			El dosier fantasma

